 Basic about order statistics

Anyone who studies the theory of statistics or uses data is familiar with the idea of ave​rage. Also, many printouts show the median and perhaps the first and third quartile (the median is the second quartile) and often the smallest and the largest values. 

While the average is a linear combination of the data (we add all the values and divide by ’n’) the median, the minimum, the maximum etc are called order statistics. This means that the dataset is ordered and thus there will be e.g. a smallest and a largest value. (In the theory below we might also talk about the second smallest or the second largest and investigate its features.) 

The macro %MinMax shows some ideas and the document Fastest scorer.doc treats also these features. In this document the two expressions F(x) and f(x) are used. See the graph in appendix for a closer explanation.
The idea with order statistics is quite natural. If we manufacture chains for example, we probably need to know about the minimum strength, as a chain is not stronger than its weakest link. When we start a number of parallel activities, where we cannot continue until all are finished, then we look at the maximum. If we start selling a large number of items we might need to think about the shortest time to failure. If we build a protection against waves we need to know something about the maximum of waves.

Below we will consider a special case. In electronics and telecommunication we transfer signals coded using the binary number system and usually we talk about ’1’ and ’0’. This is often called bits (binary digits). Suppose that we send a large number (millions) of such bits. Occasionally some of them fail and are not transferred properly. If we consider this from a distribution point of view, we would most likely say that the number of incorrect bits would follow a binomial distri​bution but as the fault rate fortunately is very low we would approximate this with the Poisson distribution.

Now, suppose that we repeatedly send a million bits (per second perhaps), count the number of faults per second and report the maximum number of bit-errors over, say, 60 seconds we have an order statistic as above. Perhaps this scheme was introduced believing that it is an efficient way of using the data instead of recording e.g. the sum.

Suppose that we have X1, X2, …, Xn independent variables (measurements). If we order them, we will write X(1), X(2), …, X(n) where X(1) is the smallest value and X(n) is the largest value. In any book that treats order statistics there is a derivation (not to difficult to follow if it is done well) of the following formula:

= the distribution function for the rth order statistic

= the distribution function for the distribution of X1, X2, …, Xn

(NB that in many books F(x) or F(r)(x) is written P(X ( x) or P(X(r) ( x) and is read as "the probabi​lity that the variable X or X(r) gets a value equal to or less than x". The first form is slightly shorter but the second form is perhaps more clear.)

F(r)(x) reads ’the probability of the rth order statistic getting a value less or equal to x’ and as soon as we know this, we can obtain and draw the distribution, calculate mean and standard devi​ation etc. The expression is often called ’the cumulative distribution function’ and is often abbrevi​ated CDF in books and computer programs. The formula says that in order to calculate F(r) (x) for each x-value, we need to calculate 

 and then do the summation. 

The formula is valid for continuous as well as discrete variables. Before discussing the best way to calculate the results we show a theoretical example and a corresponding simulation.

	An example

Suppose that we get n = 5 values from a Poisson distribution and that we sort the values from the lowest to the highest (the lowest in each sample is shown in bold):


Sample 1:
2
2
2
3
4


Sample 2:
0
2
2
4
7


Sample 3:
0
4
4
6
6


Sample 4:
0
2
2
2
4


etc.

Suppose that the intensity () of the Poisson variable is 2.3 and we want to know what the probability that the minimum is less than or equal to x = 1:

F(x) is the cdf for the Poisson distribution i.e. 

We then get

Suppose that the intensity () of the Poisson variable is 2.3 and we want to know what the probability of the minimum (i.e. r = 1) of n = 5 Poisson values is less than or equal to x = 1:

Now we want to simulate this situation to see if we then get approximately this answer. The following Minitab commands will create an estimate of the answer.

random 5000 c1-c5;             # Creates 5000 samples of five

poisson 2.3.                   # i.e. each row is a sample.

rmin c1-c5 c6                  # Gets the minimum of each sample.

                               # Store results in column c6.

let k5 = sum(c6 le 1)/n(c6)    # Calculates the proportion ’1’:s in

                               # column c6.

prin k5                        # Prints the result.

This simulation will probably come close to the theoretical value above, 0.8659.




Developing the expression.  The exercise in the square gives us only the result for x = 1, no other x-value. We want the distri​bution for all possible x-values therefore we need to increase the calcu​lation which we do by writing a small macro.

But first we want to rewrite the formula above in order to simplify the calculations. We do this via the following manipulations:

The first term in the ’split’ is equal to 1 because it is a sum over all possible values in a binomial distribution. If we subtract the terms from i = 0 to i = (r – 1) we get the desired answer and to do that we use some basic possibilities of Minitab.

One problem is the range of x-values. In case of the Poisson distribution we know that the range of the x-values is from 0 to infinity but we will not calculate the result for all these possibilities. We limit these by finding the probable range for the outcome of the original process. 

The following commands calculate and simulate the distribution of an order statistic. The com​mands need to be saved amongst the other macros in an ordinary textformat as ’OrderStat.mac’ using a word processor and run by using

MTB > %OrderStat                # See below for contents

Changes to the macro (i.e. another -value or different n) need to be done by reprogramming and resaving the macro. All commands (approx 2 pages) can be ‘copy/paste’ from this document.

Commands for calculation and simulation

gmacro

orderstat

erase c1-c1000                 # Clears the columns c1-c1000.

erase k1-k1000                 # Clears the constants c1-c1000.

note

mtitle "  For info: see www.ing-stat.nu, [Articles], 'Order statistics.doc'"

note

name c1 'x' c2 'Fr(x)' c3 'distr' c6 'Sim distr'

name k1 'Lambda'  k2 'r'  k3 'n'  k51 'Lower end of x-range' 

name k52 'Upper end of x-range'  k54 'F(x)'

let k1 = 2.3                   # Lambda, intensity of the Po-distr.

let k3 = 5                     # Number of values 'n' (max 40).

let k2 = 5                     # r-value: r = 1 is 'min', r = n is 'max'.

if k2 > k3                     # Stops if parameters are incorrect.

note

note r > n. Impossible! Execution stops.

note

exit

endif

if k3 > 40                     # Stops if parameters are incorrect.

note

note n > 40, execution stops.

note

exit

endif

invcdf 0.0001 k51;             # Lower end of x-range.

poisson k1.

invcdf 0.9999 k52;             # Upper end of x-range.

poisson k1.

let k10 = 1                    # Sets counter to one.

do k53 = k51:k52               # A do-loop for all x-values.

cdf k53 k54;                   # The 'F(x)'-value in the formula.

poisson k1.

let k55 = 'r' - 1              # Upper limit of sum.

cdf k55 k56;                   # Calculates the cdf.

bino k3 k54.

let k57 = 1 - k56              # 'F(r)(x)'-value for the given x-value. 

let c1(k10) = k53              # Stores the x-value in column c1.

let c2(k10) = k57              # Stores the 'F(r)(x)'-result in c2.  

let k10 = k10 + 1              # Row counter.

enddo

diff c2 c3                     # Creates the distribution from 'F(r)(x)'.

let c3(1) = c2(1)              # Puts value 1 in c3.

copy c1-c3 c1-c3;              # Removes very small probabilities.

omit c3 = 0:0.0001.

note

note

note ----------------------------------------------------------------

note  Column c3 now contains the probability distribution for the

note  chosen order statistic.

note

note  The next step is the simulation of the result. It might take a

note  few seconds...

note ----------------------------------------------------------------

note

let k11  = 50000               # Number of simulated values.
let k12 = 50 + k3              # Upper column number for sim. values.

random k11 c51-ck12;           # Simulates k11 Po-values in column c51-ck12

poisson k1.                    # using lamba-value k1.

    if k2 = 1                  # i.e. the minimum is to be calculated.

rmin c51-ck12 c100             # Store min (rowwise) in c100.

elseif k2 = 'n'                # i.e. the maximum is to be calculated.

rmax c51-ck12 c100             # Store max (rowwise) in c100.

else

erase c4

note

note     No simulation!

note

exit

endif

tally c100;                    # Makes a simple tally of data in c100

store c101 c102.               # and stores the result in c101 and c102.

let c5 = c101                  # Copies c101 to c5.

let c6 = c102/n(c100)          # Calculates simulated result.

copy c5 c6 c5 c6;              # Removes very small probabilities.

omit c6 = 0:0.0001.

note ----------------------------------------------------------------

note  The simulated result in c6 is probably very close to the theo-

note  retical result in c3.

note

note  Knowing the x-values (c1) and the theoretical distribution (c3)

note  we can, using ordinary mathematical tools, calculate the mean

note  value and the standard deviation for the order statistic.

note  

note ----------------------------------------------------------------

note

endmacro

The distribution of minimum

Suppose that we have during a relatively short period of time sold a number of electronic units, perhaps after a release of a new product. We suppose that the life lengths of the units is exponen​tially distributed with the parameter value  = 0.02, which gives the expected life time of 50 time units. (The macro %Random shows a picture of the distribution.) 

When can we expect the first failure, the first trouble report, the first phone call? Here we are thus interested in the minimum time value of n started units. This problem is the same problem as described in the document Fastest scorer.doc. 

Using expressions and formulas in this document we get the following. (We know, perhaps from the literature, that the cumulative distribution function F(x) for the exponential distribution is):

And this last expression is the cumulative distribution function for an exponental distribution. Thus the minimum of a number of values, where the values are exponentially distributed, is again an exponential distribution but with different parameter value namely n. Let us simulate this using  = 0.02 and n = 50:

erase c1-c100                  # Erases the first 100 columns.

let k1 = 1/0.02                # Inverts the value for Minitab.

random 1000 c1-c50;            # Simulates 1000 values in c1-c50 from

expo k1.                       # an exponential distribution.

rmin c1-c50 c51                # row min of c1-c50, store in c51.

describe c51                   # Describes column c51.

We expect the mean to be approximately    

A study of the printout reveals that the average minimum (column c51) is approximately 1. From the theory of the exponential distribution we also know that the standard deviation is equal to its mean value. Therefore the printout should show a standard deviation close to 1.

Alternative way of calculation

There is another relation that can be used, namely the relation between binomial sums and the so-called incomplete beta function. This gives us the following expression:

This new expression looks just as complicated as the previous one and at least we have a chance to calculate the previous one by hand. However, this expression happens to be the distribution func​tion (cdf) of the so-called Beta distribution, which also is available in Minitab. The Beta distribu​tion can only take values in the range [0, 1] and is often used when regarding the fault rate as a variable. Its shape can vary greatly depending on the values of the parameters a and b. (See the menus [Calc]>[Probability Distributions]>[Beta…])

Conclusion

There are many ways that the use of the order statistic, i.e. the use of minimum, median, maxi​mum, etc value in order to follow a process. However, sometimes we see the use of minimum or maximum instead of average where the ave​rage would be a better statistic as it has a lower variation. The following simulation commands show this:

erase c1-c1000                 # Erases the first 1000 columns.

random 1000 c1-c10;            # Simulates 1000 values in c1-c10 from

poisson 4.6.                   # a Poisson distribution.

name c11 'mini' c12 'aver' c13 'maxi'

rmin c1-c10 c11                # row min of c1-c10, store in c11.

rmea c1-c10 c12                # row average of c1-c10, store in c11.

rmax c1-c10 c13                # row max of c1-c10, store in c11.

descr c11-c13                  # Describes c11-c13, i.e. min, ave, max.

stack c11-c13 c14              # Stacks c11 on c12 on c13, res in c14.

set c15                        # Stores 1000 1’s, 1000 2’s and 

(1:3)1000                      # 1000 3’s in c15 for later plotting.

end

plot c14*c15;                  # Plots ‘min, average, max’.

graph;

type 1;

color 46;

Scale 1;

label 'min' 'average' 'max';

tsize 0.8;

ticks 1 2 3;

HDisplay 0 0 0 0;

Scale 2;

HDisplay 0 0 0 0;

nodt;

title 'Min, ave, max of 1000 Po-samples, each sample with n = 10';

tsize 1.1;

tcolor 4;

symb;

size 0.6;

axlabel 1 '';

axlabel 2 'Values from a Poisson distribution';

tsize 0.8.

Appendix

The two diagrams below show the distribution function (top) and the probability density distri​bution (bottom) respectively for an ordinary normal distribution. The distribution function is sometimes abbreviated as ‘cdf’ and often designated as F(x). The density distribution is often called ‘pdf’ and designated as f(x).

The mathematical relationships between the two entities is:

The top diagram: 
F(x)

The bottom diagram: 
f(x)

The size of the shaded (red) area is indicated on the Y-axis of the cdf (top graph).

